

In de BIT 90 kennen we 2 tabellen: de eerste is de interne "TOKEN" tabel, welke de standaard "TOKEN" tabel is die niet veranderd kan worden en de tweede is de externe "TOKEN" tabel die door de gebruiker veranderd kan worden.

Het startadres van de externe "TOKEN" tabel (TOKENE) wordt bijgehouden in het RAM geheugen. Wanneer de BIT 90 wordt aangezet wordt het startadres ingesteld op de interne "TOKEN" tabel. De BIT 90 zoekt naar bijpassende gereserveerde woorden van de externe en dan de interne tabel.

Hierdoor is de gebruiker in staat nieuwe commando's te creëren of bestaande commando's te wijzigen door TOKENE te veranderen.

De "TOKEN" tabel is ingedeeld zoals onderstaand:

| byte telling van (TOKENE) | gebruik |
|---------------------------|----------------------|
| 1....N | gereserveerd woord 1 |
| N+1 | token code 1 |
| N+2,N+3 | toegangspunt 1 |
| W+4m | gereserveerd woord 2 |
| M+1 | TOKEN code 2 |
| M+2, M+3 | toegangspunt 2 |
| . | . |
| . | . |
| . | . |
| laatste byte | OFFH |

TOKENE : 2 BYTES pointer in locatie 28677,28678
gereserveerd woord: Standaard ASCII string
Token code : 1 byte code met MSB (bit 7) ingesteld
Toegangspunt : Startadres van de commando processor
Deze commandoprocessor wordt aangeroepen door de BIT 90.
BASIC interpreter:
toegang: de HL registers wijzen naar de geheugen locaties volgend op die van het gereserveerd woord.
uit : de HL registers wijzen naar de eind van het statement, b.v.0 of ":"

Om uw eigen BASIC commando's te creëren, worden de volgende stappen aanbevolen:

- Stap 1 Commando processoren laden
- Stap 2 Token tabel aanmaken
- Stap 3 Wijzig de pointer van de externe "token" tabel

Bijvoorbeeld om het LIST en SAVE commando uit te schakelen:

stap 1 Laadt de commando processor: in dit geval wordt slechts 1 assembly instructie (RET, op code 201) gebruikt op adres 7500 H.

stap 2 Vervaardig de externe "TOKEN" tabel op startadres 7501H (in assembly)

```

DEFM "SAVE" ; ASCII string
DEFB 129 ; token code van "SAVE"
DEFW 7500H ; toegangspunt van de commando processor
29952

```

```

DEFM "LIST" ; ASCII string
DEFB 156 ; token code
DEFW 7500H ; toegangspunt van de commando
processor
DEFB OFFH ; einde tabel

```

Stap 3 wijzig de inhoud van geheugen locaties 7005, 7006 naar 7501H. 28677 28678

Ondergebracht in een BASIC programma ziet het er als volgt uit:

```

> LIST
10 FOR A=29952 TO 29967
20 READ D:POKE A,D
30 NEXT A
40 POKE 28677,1:POKE 28678,117
50 DATA 201,83,65,86,69,129,0,117,76,73,83,84,156,0,117,255
> RUN

```

```

READY
> LIST
READY
> SAVE
READY
>

```

READY

Voor sommige besturingskaarten van randapparaten wordt de externe "TOKEN" tabel automatisch ingesteld tijdens de koude start. (Aanzetten van de voeding). Deze tabel wordt in het RAM geheugen geschreven vanaf MEMT. Als bijvoorbeeld een Centronics Interface kaart wordt gebruikt, wordt een "token" tabel van 33 bytes in het bovenste stukje RAM opgebouwd. De nieuwe 4 BASIC commando's zijn nu beschikbaar voor programmering:

PRTSC, LLIST, LPRINT, COPY.

Om uw eigen commando's als standaard in te brengen wordt de volgende algoritme aanbevolen:

(E= pointer van de externe token tabel, I= 14854, pointer van de interne token tabel, C=2 byte mode voor controle op dubbelgebruik)

1. Als E=I THEN maak externe token tabel met eindcode OFFH, wijzig MEMT en HIMEM, wijzig nieuwe E naar C.
2. If E<>I AND E<>C THEN maak dan een externe token tabel zonder eindcode, wijzig MEM en HIMEM, wijzig nieuwe E naar C.
3. Executeer het programma met nieuwe commando's.

Het volgende BASIC programma wordt gebruikt als TIMER en geeft de uren, minuten en seconden weer in het midden van het scherm. Een nieuw commando LOCATE wordt aangemaakt voor cursorbesturing. De syntax is "LOCATE Y,X", waarbij Y ligt tussen 0 en 23 en X tussen 0 en 29. Als Y of X groter is dan 255 wordt de boodschap "SYNTAX ERROR" op het scherm afgedrukt. Let op dat de programma regels 10 tot 225 eerst moeten worden geexecuteerd en dan pas de nieuwe statements uitgevoerd. Regel 230 vervaardigt een interrupt van een 1/50 seconde; regel 240 wacht 1 seconde;

regel 250 telt de uren, minuten en seconden; regel 260 gebruikt LOCATE" om de informatie op het scherm zichtbaar te maken. De gebruikte geheugenlocaties zijn als volgt:

| adres | omschrijving |
|--------------|---|
| 29952, 29953 | 2 bytes controle code |
| 28677, 28678 | pointer van de externe token tabel (TOKENS) |
| 29160, 29161 | laatste adres van het vrije RAM-geheugen MEMT |
| 29162, 29163 | laatste adres van het vrije RAM-geheugen HIMEM |
| 29954 | toegangspunt in de commando processor voor "LOCATE" |

```

10 C=PEEK( 29953)*256+PEEK( 29952)
20 E=PEEK( 28678)*256+PEEK( 28677)
30 I=14854:MT=PEEK( 29161)*256+PEEK( 29160)
40 IF E<> 14854 AND E<>C THEN GOTO 70
50 IF E=14854 THEN POKE MT,255:MT=MT-1:GOTO 70
60 GOTO 230
70 FOR A=29954 TO 29994
80 READ D:POKE A,D
90 NEXT A
100 FOR A=MT-8 TO MT
110 READ D:POKE A,D
120 NEXT A
130 MT=MT-9
140 POKE 29161,MT/256:POKE 29160,MT-INT( MT/256)*256
150 POKE 29163,PEEK( 29161):POKE 29162,PEEK( 29160)
160 POKE 28677,PEEK( 29160)+1:POKE 28678,PEEK( 29161)
170 POKE 29952,PEEK( 28677):POKE 29953,PEEK( 28678)
180 DATA 205,64,37,245,126,254,44,194,243,2
190 DATA 35,205,64,37,209,60,60,230,31,95
200 DATA 175,203,26,31,203,26,31,203,26,31
210 DATA 179,95,122,230,3,87,237,83,15,112,201
220 DATA 76,79,67,65,84,69,224,2,117
230 HOME :MUSIC 1,↑,*, "R":PLAY
240 IF PEEK( 29532)<50 THEN GOTO 240
250 POKE 29532,0:SC=SC+1:IF SC=60 THEN MI=MI+1:SC=0:IF MI=60 THEN
    HR=HR+1:MI=0: IF HR=24 THEN HR=0
260 LOCATE 12,11:PRINT SPC( 10):LOCATE 12,11:PRINT HR;" ";MI;" ";
    ;SC;
270 GOTO 240

```